NAG Fortran Library Routine Document

F01QJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F01QJF finds the RQ factorization of the real m by n $(m \leq n)$ matrix A, so that A is reduced to upper triangular form by means of orthogonal transformations from the right.

2 Specification

```
SUBROUTINE F01QJF(M, N, A, LDA, ZETA, IFAIL)
INTEGER            M, N, LDA, IFAIL
real               A(LDA,*), ZETA(*)
```

3 Description

The m by n matrix A is factorized as

$$A = (\, R \quad 0 \,) P^{\mathrm{T}} \quad \text{when } m < n,$$

$$A = RP^{\mathrm{T}} \quad \text{when } m = n,$$

where P is an n by n orthogonal matrix and R is an m by m upper triangular matrix.

P is given as a sequence of Householder transformation matrices

$$P = P_m \dots P_2 P_1,$$

the $(m - k + 1)$th transformation matrix, P_k, being used to introduce zeros into the kth row of A. P_k has the form

$$P_k = I - u_k u_k^{\mathrm{T}},$$

where

$$u_k = \begin{pmatrix} w_k \\ \zeta_k \\ 0 \\ z_k \end{pmatrix},$$

ζ_k is a scalar, w_k is an $(k - 1)$ element vector and z_k is an $(n - m)$ element vector. u_k is chosen to annihilate the elements in the kth row of A.

The vector u_k is returned in the kth element of ZETA and in the kth row of A, such that ζ_k is in ZETA(k), the elements of w_k are in A$(k, 1), \dots ,$ A$(k, k - 1)$ and the elements of z_k are in A$(k, m + 1), \dots ,$ A(k, n). The elements of R are returned in the upper triangular part of A.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, Oxford

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of A.

When M=0 an immediate return is effected.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of A.

Constraint: $\mathrm{N} \geq \mathrm{M}$.

3: A(LDA,*) – ***real*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the leading m by n part of the array A must contain the matrix to be factorized.

On exit: the m by m upper triangular part of A will contain the upper triangular matrix R, and the m by m strictly lower triangular part of A and the m by $(n-m)$ rectangular part of A to the right of the upper triangular part will contain details of the factorization as described in Section 3.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F01QJF is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{M})$.

5: ZETA(*) – ***real*** array *Output*

Note: the dimension of the array ZETA must be at least $\max(1, \mathrm{M})$.

On exit: ZETA(k) contains the scalar ζ_k for the $(m-k+1)$th transformation. If $P_k = I$ then ZETA(k) = 0.0, otherwise ZETA(k) contains ζ_k as described in Section 3 and ζ_k is always in the range $(1.0, \sqrt{2.0})$.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= -1$

On entry, $\mathrm{M} < 0$,
or $\mathrm{N} < \mathrm{M}$,
or $\mathrm{LDA} < \mathrm{M}$.

7 Accuracy

The computed factors R and P satisfy the relation

$$(R \quad 0) P^{\mathrm{T}} = A + E,$$

where

$$\|E\| \leq c\epsilon\|A\|,$$

ϵ is the ***machine precision*** (see X02AJF), c is a modest function of m and n, and $\|.\|$ denotes the spectral (two) norm.

8 Further Comments

The approximate number of floating-point operations is given by $2m^2(3n - m)/3$.

The first k rows of the orthogonal matrix P^{T} can be obtained by calling F01QKF, which overwrites the k rows of P^{T} on the first k rows of the array A. P^{T} is obtained by the call:

```
IFAIL = 0
CALL F01QKF('Separate',M,N,K,A,LDA,ZETA,WORK,IFAIL)
```

WORK must be a $\max(m - 1, k - m, 1)$ element array. If K is larger than M, then A must have been declared to have at least K rows.

Operations involving the matrix R can readily be performed by the Level 2 BLAS routines F06PJF (STRSV/DTRSV) and F06PFF (STRMV/DTRMV) (see Chapter F06), but note that no test for near singularity of R is incorporated into F06PJF (STRSV/DTRSV). If R is singular, or nearly singular then F02WUF can be used to determine the singular value decomposition of R.

9 Example

To obtain the RQ factorization of the 3 by 5 matrix

$$A = \begin{pmatrix} 2.0 & 2.0 & 1.6 & 2.0 & 1.2 \\ 2.5 & 2.5 & -0.4 & -0.5 & -0.3 \\ 2.5 & 2.5 & 2.8 & 0.5 & -2.9 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F01QJF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA
      PARAMETER        (MMAX=10,NMAX=20,LDA=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N
*     .. Local Arrays ..
      real             A(LDA,NMAX), ZETA(MMAX)
*     .. External Subroutines ..
      EXTERNAL         F01QJF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F01QJF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      WRITE (NOUT,*)
      IF ((M.GT.MMAX) .OR. (N.GT.NMAX)) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'M or N is out of range.'
```

```
         WRITE (NOUT,99999) 'M = ', M, '   N = ', N
      ELSE
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         IFAIL = 0
*
*        Find the RQ factorization of A
         CALL F01QJF(M,N,A,LDA,ZETA,IFAIL)
*
         WRITE (NOUT,*) 'RQ factorization of A'
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Vector ZETA'
         WRITE (NOUT,99998) (ZETA(I),I=1,M)
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     + 'Matrix A after factorization (R is in left-hand upper triangle)'
         DO 20 I = 1, M
            WRITE (NOUT,99998) (A(I,J),J=1,N)
   20    CONTINUE
      END IF
      STOP
*
99999 FORMAT (1X,A,I5,A,I5)
99998 FORMAT (5(1X,F8.4))
      END
```

9.2 Program Data

```
F01QJF Example Program Data
  3     5                       :Values of M and N
  2.0   2.0   1.6   2.0   1.2
  2.5   2.5  -0.4  -0.5  -0.3
  2.5   2.5   2.8   0.5  -2.9   :End of matrix A
```

9.3 Program Results

```
 F01QJF Example Program Results

 RQ factorization of A

 Vector ZETA
   1.0092   1.2981   1.2329

 Matrix A after factorization (R is in left-hand upper triangle)
  -3.1446  -1.0705  -2.2283   0.6333   0.7619
   0.5277  -2.8345  -2.2283  -0.1662   0.0945
   0.3766   0.3766  -5.3852   0.0753  -0.4368
```